Subject: Re: Reading Multiple DICOM Files Posted by rtk on Mon, 09 Mar 2009 18:07:55 GMT

View Forum Message <> Reply to Message

```
On Mar 8, 6:17 pm, Jye <jye.sm...@gmail.com> wrote:

> if i eq 0 then All_Images = image_slice else All_Images =

> [[[All_Images]], [[image_slice]]]
```

Forget too much Java, which, I agree, is a bad thing. I'd say you have been doing too much Matlab programming. The vast majority of your problem lies with the line above. If I run this code on my box:

```
pro ttt0
  s = systime(1)
  for i=0,899 do begin
     im = fix(dist(256))
     if (i eq 0) then begin
        all = im
     endif else begin
        all = [[[all]], [[im]]]
     endelse
  endfor
  print, 'runtime = ', systime(1)-s
  help, all
end
It takes 93 seconds. However, if I run this code:
pro ttt2
  all = intarr(256,256,900)
  s = systime(1)
  for i=0,899 do begin
     im = fix(dist(256))
     all[*,*,i] = im
  endfor
  print, 'runtime = ', systime(1)-s
  help, all
end
```

it takes 1.6 seconds. The only difference is that the output array is allocated once, instead of 900 times with 899 ever-increasing array copies to boot. Since you know the number of file in advance, allocate All_Images and just assign into it.

Ron